
Abstract 

SAS® software is the “gold” standard for robust and reliable data access, manipulation, analytics, analysis, reporting, and data 

discovery. Microsoft Excel® is the most widely used software in the world. This paper and presentation introduces the Output 

Delivery System (ODS) Excel destination; and three great reasons to use the SAS ODS Excel destination; along with the 

application and construction of SAS and ODS Excel code to share data, tables, statistics, images, reports, and results to create 

exciting and robust Microsoft Excel files and workbooks. 

Introduction 

Technology is moving at an astounding pace. With many new features and enhancements introduced in Version 9.4, users can 

finally sound the trumpets for what could be the most exciting time for output delivery since the introduction of the color 

monitor and printer. Never again will SAS users be confined to boring monospace output. Instead, output delivery has entered 

a new age taking full advantage of font characteristics, color, a variety of output layouts, and numerous other features. This 

paper and presentation will provide users with many new and exciting features available with the ODS Excel destination. 

Sprinkled throughout will be numerous tips, tricks, and techniques that will, hopefully, make the learning process a little easier. 

 

Attendees will learn numerous tips, tricks, and techniques in handling output including the advantages of using the ODS Excel 

destination; how ODS handles raw data; how ODS combines raw data with table definitions; how to open, close, and manage 

the ODS Excel destination; how the Results window stores links to ODS output; how selection and exclusion lists are used; and 

how to send and/or create Excel spreadsheets from SAS data sets and results using the ODS Excel destination. 

 

Table Used in Examples 

The data set used in all the examples in this paper is the SASHELP.CARS. The SASHELP.CARS data set consists of 428 

observations and 15 variables, illustrated below. 

Output Delivery Basics 

Years ago, ODS was released to the user community as a way to address the inherent weaknesses found with the creation of 

traditional reports and output results. ODS enables “quality” looking data, reports, and output results to be produced without 

having to import, or use copy and paste techniques, into word processors such as MS-Word. Although the SAS® Output Delivery 

System (ODS) continues to support the creation of traditional SAS Listing or monospace output (i.e., Listing), it provides users 

with many ways to access and format data, reports, and output results. ODS provides many new features providing users with 

greater flexibility including the Output, RTF, PDF, HTML, XML, PowerPoint, and Excel destinations. In Version 9.4 (M3), many 

new output formatting features and options are introduced for SAS users to take advantage of, including the ODS Excel 

destination. Users have a powerful and easy way to create and access formatted procedure and DATA step output. 

Tip #1 – ODS and “Batch” Use 

Many of the ODS features found in the interactive side of the SAS Display Manager System (DMS) can also be used in batch 

processing. ODS has been designed to make exciting new formatting options available to users. In a windowing environment, 

ODS can send output to the following destinations: the output window (DMS), the listing file, HTML, SAS dataset, rich text 

format (RTF), postscript file, external output file (non-SAS file), or output device. The only exception for batch processing is 

having output sent to the output window. 

Tip #2 – What if I’m Still Not Using the latest Version of SAS Software 

First introduced in Version 6.12, ODS offered users the capability to format output to destinations other than traditional line 

printers. Version 9.4 (M3) introduced the ODS Excel destination to deploy output to the web, the creation of SAS datasets and 

rich text format (RTF) files, and DATA step interaction. ODS was designed to address the inherent weaknesses found in 

traditional SAS output. It enables the creation of “quality” looking output without having to import it into word processors such 

as MS-Word. New output enhancements were introduced in Version 8 and then in Version 9, including the ability to create 

postscript files and output customizations. To take full advantage of the power offered in ODS, it is recommended that users 

upgrade to the latest Version as early as possible to take advantage of these features. 

Tip #3 – ODS and System Resources 

A very important efficiency consideration is to remember that ODS currently supports the following destinations: 1) Listing, 2) 

rich text format (RTF), 3) postscript, 4) HTML, 5) PowerPoint, 6) Excel and 7) Output. (Note: It also provides support with the 

creation of output and results in the DATA step.) Each ODS destination can be open or closed at the same time. For each open 

destination, ODS sends output object(s) to it. System resources are used when a destination is open. As a result, make sure any 

and all unwanted open destinations are closed to conserve on resources. 

Tip #4 – Closing Destinations before and after use 

The Listing destination is open by default at SAS invocation, while the other destinations are closed. If nothing is done to 

suppress output to the Listing destination, your SAS programs automatically produce Listing output, just as they always have in 

the SAS System. If you needed to suppress printed output from being sent to the Listing destination (or DMS Output window) 

before the execution of a procedure step, the following ODS statement would be issued: 

 

ODS Listing Close; 

  Proc univariate data=sashelp.Cars; 

  Run; 

ODS Listing; 

 

By closing the Listing destination before the procedure code, the SAS System is actually suppressing output to that destination 

until it is reopened. The preceding example shows that at the end of the procedure step, the Listing destination is reopened by 

specifying ODS Listing; so output from subsequent steps can be sent to the Listing destination. 

Tip #5 – Deleting Output from the Results Window 

The Results window identifies procedure output that has been produced, providing users with an improved way to manage 

their output. It is customarily a good thing to remove unwanted output displayed in this window to conserve on system 

resources. The Results window is opened by specifying the command ODSRESULTS on the DMS command line or by selecting 

View Results from the pull-down menu. 

 

To delete procedure output, use the following steps: 

 

1. Select the procedure folder you want to remove. 

2. Click the Delete button on the task bar. 

3. Select “Yes” to confirm the deletion of the procedure output folder. 

 

 

 

Tracing Output 

Output producing procedures create one or more pieces or tables of information. In order to discriminate between the various 

pieces of information, it is advantageous to know the names assigned to each piece of information. The ability to display the 

names of individual pieces of information generated on output is referred to as tracing. 

 

Tip #6 – Tracing Procedure Output 

The trace record displays information about the data component, the table definition, and the output object. By specifying an 

ODS Trace ON ; statement, the SAS System turns the trace feature on and prints results to the SAS Log destination. 

 

ODS Trace ON ; 

  Proc univariate data=sashelp.Cars ; 

  Run ; 

ODS Trace Off ; 

 

For example, the trace record displays the following output objects to the SAS Listing destination: 1) Moments, 2) 

BasicMeasures, 3) TestForLocation, 4) Quantiles, and 5) ExtremeObs. A sample trace record containing each output object’s 

name, label, template, and path is displayed for the Univariate procedure. Note that for each output object, the name, label, 

template, and path is displayed. 

 

 Output Added: 
 ------------- 
 Name:       Moments 
 Label:      Moments 
 Template:   base.univariate.Moments 
 Path:       Univariate.MSRP.Moments 
 ------------- 
  
 Output Added: 
 ------------- 
 Name:       BasicMeasures 
 Label:      Basic Measures of Location and Variability 
 Template:   base.univariate.Measures 
 Path:       Univariate.MSRP.BasicMeasures 
 ------------- 
  
 Output Added: 
 ------------- 
 Name:       TestsForLocation 
 Label:      Tests For Location 
 Template:   base.univariate.Location 
 Path:       Univariate.MSRP.TestsForLocation 
 ------------- 

 Output Added: 
 ------------- 
 Name:       Quantiles 
 Label:      Quantiles 
 Template:   base.univariate.Quantiles 
 Path:       Univariate.MSRP.Quantiles 
 ------------- 
  
 Output Added: 
 ------------- 
 Name:       ExtremeObs 
 Label:      Extreme Observations 
 Template:   base.univariate.ExtObs 
 Path:       Univariate.MSRP.ExtremeObs 
 ------------- 

 
 

Selecting Output with ODS 

A selection or exclusion list exists for each open ODS destination. These lists determine which output objects to send to ODS 

destinations. To accomplish this, ODS verifies whether an output object is included in a destination’s selection or exclusion list. 

If it does not appear in this list, then the output object is not sent to the ODS destination. If an output object is included in the 

list, ODS determines if the object is included in the overall list. If it does not appear in this list, then the output object is not sent 

to the ODS destination. If an output object is included in the overall list then ODS sends it to the selected destination. 

 

Tip #7 – Selecting Desired Pieces of Information 

Once you know the individual names of each output object (from the trace), you can then select the desired object for 

reporting purposes. The syntax is: 

 

 ODS select output-component-name; 

 

where output-component-name is the name of the desired output object. To select just the output object Moments from the 

Univariate procedure, the following syntax is specified: 

 

ODS Select Moments ; 

Proc univariate data=sashelp.Cars ; 

Run ; 

 

 

Export, Import and the LIBNAME Engine 

Creating and accessing Excel files requires a SAS/ACCESS to PC Files license. SAS/ACCESS includes a component called PC Files 

Server. It accepts requests from SAS to convert to and from Excel (and other file types), allows newer 64-bit desktop machines 

to work with older 32-bit machines, and circumvents the bit architecture mismatch. 

 

Using PROC EXPORT 

PROC EXPORT reads data from a SAS data set and writes it to an external data source. A variety, but incomplete list, of external 

data sources include: 

 

 MS-Access databases 

 MS-Excel files 

 SPSS files 

 Stata files – rectangular files stored in memory (Reference: http://data.princeton.edu/stata/DataManagement.html) 

 Delimited external files 

 Lotus 1-2-3 spreadsheet files 

 Paradox files 

 

Tip #8 – Creating an Excel File Under Windows with PROC EXPORT 

Creating an Excel file is easy with PROC EXPORT. To create an Excel spreadsheet under Windows from the SASHELP.CARS data 

set, specify the following statement and options: 

 

 PROC EXPORT DATA=sashelp.Cars 

          OUTFILE='c:\Cars.xls' 

             DBMS=EXCEL 

             REPLACE; 

 RUN; 

 

Tip #9 – Creating an Excel File Under Unix with PROC EXPORT 

To create an Excel file under Unix from the SASHELP.CARS data set, specify the following statement and options: 

 

PROC EXPORT DATA=sashelp.Cars 

         OUTFILE='c:\Cars.xls' 

            DBMS=EXCELCS 

            REPLACE; 

RUN; 

 

Tip #10 – Creating a Subsetted Excel File with PROC EXPORT 

Exporting a subset to an Excel file is accomplished using PROC EXPORT with a WHERE= data set option. For example, to export 

the “SUV” vehicles from the SASHELP.CARS data set to an Excel file, specify the following PROC EXPORT statement and WHERE= 

data set option: 

 

PROC EXPORT DATA=SASHELP.CARS 

          (WHERE=(type=“SUV”)) 

         OUTFILE='c:\SUV_Vehicles.xls' 

            DBMS=EXCEL; 

RUN; 

 

 

Using PROC IMPORT 

PROC IMPORT reads data from an Excel file to a SAS data set. A variety, but incomplete list, of a variety of external data sources 

include: 

 

 MS-Excel files 

 MS-Access databases 

 Lotus 1-2-3 spreadsheet files 

 

Tip #11 – Creating a SAS Data Set Under Windows with PROC IMPORT 

To create a SAS data set from an Excel file under Windows from the SASHELP.CARS data set, specify the following PROC IMPORT 

statement and options: 

 

PROC IMPORT OUT=sashelp.Cars 

       DATAFILE='c:\Cars.xls' 

           DBMS=EXCEL2002 

           REPLACE ; 

RUN ; 

PROC PRINT DATA=sashelp.Cars noobs ; 

RUN ; 

Tip #12 – Creating a SAS Data Set Under Unix with PROC IMPORT 

This example illustrates the syntax used to create a SAS data set from an Excel file under Unix. 

 

PROC IMPORT OUT=sashelp.Cars 

       DATAFILE='c:\Cars.xls' 

           DBMS=EXCELCS 

           REPLACE ; 

RUN ; 

PROC PRINT DATA=sashelp.Cars noobs ; 

RUN ; 

 

 

Using the LIBNAME Engine 

The LIBNAME engine associates a SAS libref with a Microsoft Excel workbook or Microsoft Access database. The LIBNAME 

syntax to create a Windows Excel file is: 

 

LIBNAME libref EXCEL <physical-file-name> 

         <SAS/ACCESS-engine-connection-options> 

         <SAS/ACCESS-libname-options>; 

 

Where 

 

     libref represents the user-defined alias to reference a SAS data set, 

     EXCEL is the SAS/ACCESS name for the interface to PC files under Windows, 

     SAS/ACCESS-engine-connection-options provides information to connect to your PC files, 

     SAS/ACCESS-libname-options defines how SAS is to interact with your data source. 

 

 

Tip #13 – Exporting a SAS Data Set to a Windows-based Excel File 

A LIBNAME statement can be used to export a SAS data set to a Windows-based Excel file. For example, to export the data in 

the SASHELP.CARS data set to a Windows Excel file, specify the following LIBNAME statement and XLS options: 

 

LIBNAME MYXLS EXCEL “c:/Cars.XLS” ; 

  DATA MYXLS.Sheet1 ; 

    SET sashelp.Cars ; 

  RUN ; 

LIBNAME MYXLS CLEAR ; 

 

Tip #14 – Exporting a SAS Data Set to a Unix-based Excel File 

A LIBNAME statement can be used to export a data set to a Unix-based Excel file. For example, to export the data in the 

SASHELP.Cars data set to an Excel file, specify the following LIBNAME statement and XLS options: 

 

LIBNAME MYXLS PCFILES PATH=“c:/Cars.XLS” ; 

   DATA MYXLS.Sheet1 ; 

     SET sashelp.Cars ; 

   RUN ; 

LIBNAME MYXLS CLEAR ; 

Creating Unique Output with ODS 

Output Delivery System (ODS) can be used to create a variety of output formats. ODS statements are classified as global 

statements and are processed immediately by the SAS System. ODS statement options control what format engine(s) are 

turned on and in effect during the step or until another ODS statement is specified. ODS has built-in format engines (e.g., 

Listing, Output, RTF, PDF, DATA Step, HTML, Excel, PowerPoint and XML). Specifying an ODS statement and destination at a 

particular point in a program is important, because output-producing PROC and DATA steps will respond by sending output to 

the open destination. 

 

 

 

 

Creating SAS Output Data Sets 

Occasionally, output results are needed in a SAS data set rather than in printed form such as the Listing destination. Re-

directing SAS procedure output to a data set is relatively simple with ODS. The syntax is: 

 

ODS Output output-table-name = user-defined-table-name; 

    < SAS Code > 

 

where output-table-name is the name of the desired output table (component) containing the information you want written to 

a data set, such as Moments in the UNIVARIATE procedure. User-defined-table-name is the name you supply for the newly 

created data set. 

 

Tip #15 – SAS Output Data Sets 

A data set can be defined as either a temporary or permanent data set. Once an object is selected, specify the object in the ODS 

OUTPUT statement. For example, the Moments from the Univariate procedure is selected and output to a SAS data set in the 

following code. 

 

ODS Listing Close ; 

ODS Output Moments = Cars_Moments ; 

Proc univariate data=sashelp.Cars ; 

Run ; 

ODS Listing; 

 

When the OUTPUT destination is no longer needed, it can be closed with the following ODS statement: 

 

 ODS OUTPUT CLOSE; 

Tip #16 – Creating Rich Text Format (RTF) with ODS 

Rich text format (RTF) is text consisting of formatting attributes codes, such as boldface, italics, underline, etc. It is principally 

used to encapsulate text and formatting attributes during copy-and-paste operations. Because word-processing programs use 

RTF rather than ASCII when handling data, the need to reformat is a thing of the past. The syntax to create RTF output is: 

 

ODS RTF FILE = ‘user-specified-file-name’; 

 

where user-specified-file-name references a complete and fully-qualified output location for the creation and storage of the 

RTF file, data, and codes. For example, the following code creates an RTF file using the Univariate procedure output. (Note: The 

RTF extension is required). 

 

ODS Select Moments ; 

ODS RTF FILE=’RTF-univariate-output.rtf’ ; 

  Proc univariate data=sashelp.Cars ; 

    Title1 ‘Delivering RTF Output’ ; 

  Run ; 

ODS RTF Close ; 

 

Two RTF output objects are produced and displayed below: 

 

 
 

 
 

 
 

Tip #17 – Creating and Naming Postscript Files 

ODS enables the creation of a postscript file from SAS output to help ensure that formatting and content is preserved while 

maintaining complete printer independence. The next example shows the creation of a postscript file from Print procedure 

output. 

ODS Printer Postscript ; 

Proc print data=sashelp.Cars noobs n ; 

  Title1 ‘Example Creating a Postscript File’ ; 

Run ; 

ODS Printer Close ; 

 

Since the postscript file was created without assigning a filename, ODS assigns the name: SASPRT.PS.  Frequently, a user-

defined filename is desired rather than the default name. In these cases, the ODS statement and File= option can be used to 

assign a name to the postscript file. 

 

ODS Printer Postscript File=’Cars’ ; 

  Proc print data=sashelp.Cars noobs n ; 

    Title1 ‘Example Creating a Postscript File’ ; 

  Run ; 

ODS Printer Close ; 

 

 

Tip #18 – The Custom Reporting Interface – ODS and the DATA Step 

To provide greater capabilities when working with custom output created in the DATA step, ODS is integrated into the DATA 

step. Two new options are necessary to take advantage of ODS: 1) the ODS option in the FILE statement and 2) the _ODS_ 

option in the PUT statement. These two options are used to direct the results of a DATA step to ODS. For example, the 

following code directs output from a DATA step through the RTF format engine to create a RTF file. 

 

ODS RTF FILE=’ods-DATA-step.rtf’ ; 

  Title1 ‘Great Cars’ ; 

  Data NULL ; 

    Set sashelp.Cars(where=(type=”SUV”)) ; 

    File print ods ; 

    Put _ods_ ; 

  Run ; 

ODS RTF Close ; 

 

The RTF file created in the previous DATA step code is illustrated below. 

Tip #19 – Available “Custom” SAS Styles 

Although many default styles that SAS uses with their output look fine, users have many more choices to choose from. To view 

the available template styles, users can submit a LIST Styles statement using PROC TEMPLATE. 

 

PROC TEMPLATE ; 

  LIST STYLES ; 

RUN ; 

 

Output and SAS Styles 
 

  

 

 

 

Tip #20 – Using the “Custom” BarrettsBlue SAS Style 

This example specifies an ODS HTML statement and a STYLE=BarrettsBlue option with PROC PRINT to send the subset of data 

from the SASHELP.CARS data set to Excel. 

 

ODS HTML FILE=‘PRINT-to-Excel.XLS’  

        STYLE=BarrettsBlue ; 

  proc print data=sashelp.Cars(where=(type=”SUV”)) noobs ; 

      title ; 

  run ; 

ODS HTML CLOSE ; 

Results 
 

 
 
 

Tip #21 – Using the “Custom” Excel SAS Style 

This example specifies an ODS HTML statement and a STYLE=Excel option with PROC PRINT to send the contents of the 

SASHELP.CARS data set to Excel. 

 

ODS HTML FILE=‘PRINT-to-Excel.XLS’  

        STYLE=Excel ; 

  proc print data=sashelp.Cars(where=(type=”SUV”)) noobs ; 

    title ; 

  run ; 

ODS HTML CLOSE ; 

 

Results 

Tip #22 – Using the “Custom” HighContrast SAS Style 

In this example, a data subset is sent to Excel using an ODS HTML statement along with an XLS extension and a STYLE= 

HighContrast option, a KEEP= data set option and a WHERE statement in a PROC PRINT. 

 

ODS HTML FILE=‘PRINT-subset-to-Excel.XLS’  

        STYLE=HighContrast ; 

  title ; 

  proc print data=sashelp.Cars(keep=make type origin msrp invoice) ; 

    Where type=”SUV” ; 

  run ;  

ODS HTML CLOSE ; 

 

Results 
 

 
 

  



Three Great Reasons to Use the SAS® ODS Excel® Destination, continued SCSUG 2017 

 
Page 13 

 

The ODS Excel Destination 

The ODS Excel Destination became production in SAS 9.4 (M3). It serves as an interface between SAS and Excel: 

 

 SAS Results and Output can be sent directly to Excel 

 Offers a Flexible way to create Excel files 

 Supports Reports, Tables, Statistics and Graphs 

 Formats Data into Excel Worksheet cells 

 Permits Automation of Production-level Workbooks 

 

The ODS Excel destination easily sends output and results to Excel. The ODS Excel syntax simplifies the process of sending 

output, reports, tables, statistics and graphs to Excel files. The ODS Excel options are able to: 

 

 Programmatically generate output and results 

 Control font sizes 

 Add special features to row and column headers 

 Adjust row and column sizes 

 Format data values 

 Align data to the left, center or right 

 Add hyperlinks for drill-down capability 

 

Tip #23 – Sending Results to an Excel File 

In this example, the PROC MEANS Statistics is sent to Excel by specifying an ODS Excel statement along with an XLSX extension. 

 

ods excel file='e:/PROCMEANS.xlsx' ; 

  proc means data=sashelp.cars ; 

    class type ; 

  run ; 

ods excel close ; 

 

Results 

Tip #24 – Sending Tabular and Graphical Results to an Excel File 

In this example, a PROC FREQ HBAR Histogram is sent to Excel by specifying an ODS Excel statement along with an XLSX 

extension. 

 

ods Excel file=“e:/Freq-HBAR-Histogram.xlsx" ; 

  PROC FREQ DATA=sashelp.cars 

           ORDER=data ; 

ods exclude twowayfreqs ; 

    TABLES Type * Origin /  

                 PLOTS=freqplot(orient=horizontal) ; 

  RUN ; 

ods Excel close ; 

 

Results 

  

 

 

Tip #25 – Sending Report Results to an Excel File with ODS Excel 

In this example, PROC REPORT sends output to Excel by specifying an ODS Excel statement. 

 

ods Excel file="PROC-REPORT.xlsx" ; 

  PROC REPORT DATA=SASHELP.CARS 

             (KEEP=Type Make Model Origin MSRP)  NOWINDOWS ; 

    WHERE MSRP < 30000 AND Type = 'Sports' ; 

    COLUMNS Type Origin Make Model MSRP ; 

    DEFINE Type   / ORDER  WIDTH=8 ; 

    DEFINE Origin / ORDER  WIDTH=6 ; 

    DEFINE Make   / ORDER  WIDTH=13 CENTER ; 

    DEFINE Model  / DISPLAY WIDTH=40 ; 

    DEFINE MSRP   / DISPLAY 'MSRP' WIDTH=8 ; 

  RUN ; 

ods Excel close ; 

Results 

 

 

 

Tip #26 – Sending Statistical and Graphical Results to an Excel File 

In this example, PROC UNIVARIATE sends output to Excel by specifying an ODS Excel statement. 

 

OPTIONS ORIENTATION=PORTRAIT ; 

ods Excel file="/folders/myfolders/Univariate-VBAR-Histogram.xlsx" ; 

  TITLE "Histogram for Chart Variable" ; 

  PROC UNIVARIATE DATA=sashelp.cars 

                 (KEEP=type origin make cylinders MSRP) 

                  NOPRINT ; 

    VAR cylinders ; 

    HISTOGRAM cylinders / NORMAL; 

  RUN ; 

ods Excel close ; 

 

Results 

Tip #27 – Creating an Excel File with the %DS2CSV Macro 

Users are able to create an Excel file using the %ds2csv macro. To create an Excel file, specify the CSV extension, runmode=b 

(batch), and the filename in the csvfile= in the macro. A list of available parameters used with the %DS2CSV macro is illustrated 

in the following table. 

 

Parameter Description 

DATA= Name of SAS data set to export. 

CSVFILE= CSV file to be created. 

RUNMODE= Specify “B” for batch or interactive SAS programs. 

OPENMODE= Specify “Replace” or “Append”. 

COLHEAD= Specify “Y” or “N” to add column headers in CSV file. 

FORMAT= Specify “Y” or “N” to keep formats. 

LABELS= Specify “Y” or “N” to keep labels on column headings. 

WHERE= Specify expression to subset data in resulting CSV file. 

 

 

In this example, the %DS2CSV macro reads the SASHELP.CARS data set and creates a CSV file called, DS2CSV. 

 

%ds2csv (data=sashelp.cars 

        ,runmode=b 

        ,csvfile=c:\mwsug\lafler\DS2CSV.csv); 

 

Log: 

NOTE: CSV file successfully generated for SASHELP.CARS. 

 

Results 

Tip #28 – Sending Statistical Results with a Background Image to an Excel File 

In this example, PROC UNIVARIATE sends output to Excel by specifying an ODS Excel statement. 

 

proc template ;  

  define style styles.background ; 

    parent=styles.excel ; 

    class body/background=_undef_ 

             backgroundimage=“e:/Analytics.jpg" ; 

  end ; 

run ; 

ods excel file="ODS-Excel-MEANS-Background.xlsx" 

         options(embedded_titles="yes" 

         start_at="5,5") 

         style=styles.background ; 

  proc means data=sashelp.cars ; 

    class type ; 

  run ; 

ods excel close ; 

 

Results 

 

 

 

Output Delivery Goes Web 

The Web offers incredible potential that impacts all corners of society. With its increasing popularity as a communications 

medium, Web publishers have arguably established the Web as the greatest medium ever created. Businesses, government 

agencies, professional associations, schools, libraries, research agencies, and a potpourri of society’s true believers have 

endorsed the Web as an efficient means of conveying their messages to the world. 

 

The Web is not a static environment, but a dynamic medium capable of distributing content anywhere and at any time. The 

24/7 model permits information to be refreshed and updated continuously as new material becomes available. A primary 

objective of Web publishers everywhere is to engage visitors with timely and interesting content that brings them back for 

more. 

Tip #29 – Pagesize / Linesize Settings 

The Options PS= and LS= have no effect when used with the HTML destination (opposed to most other output-producing steps 

that generate output to a print destination). If the PS= and/or LS= options are used with the HTML destination, they are simply 

ignored. The SAS System creates a type of “streaming” or continuous output and adds elevator bars (horizontal and/or vertical) 

for easy navigation. 

 

The SAS System does provide a way for users to paginate through output displayed in a body file. The HTML destination 

provides a way to designate an optional description of each page of the body file. The PAGE= file (when specified) recognizes 

each new page of output produced by ODS. What ODS does is create a section called Table of Pages containing links to the 

body file for easy navigation through output. 

 

Tip #30 – Deploying Output to the Web 

With the popularity of the Internet, you may find it useful to deploy selected pieces of output on your web site. ODS makes 

deploying procedure output to the web a simple process. Syntactically correct HTML code is automatically produced and made 

ready for deployment using one of the Internet browser software products (e.g., Internet Explorer, Netscape Navigator, etc.). 

As a result, the SAS System and the HTML destination create a type of “streaming” or continuous output by adding elevator 

bars (horizontal and/or vertical) for easy navigation. 

 

Tip #31 – Creating HTML Destination Files with ODS 

Four types of files can be created with the ODS HTML destination: 1) body, 2) contents, 3) page, and 4) frame. Each file is 

described below. 

 

The Body file contains the results from the procedure embedded in ODS-generated HTML code. Horizontal and vertical scroll 

bars are automatically placed on the generated page, if necessary. 

 

The Contents file consists of a link to each HTML table within the body file. It uses an anchor tag to link to each table. By using 

your browser software, you can view the contents file directly or as part of the frame file. 

 

The Page file consists of a link to each page of ODS created output. By using your browser, you can view the page file directly or 

as part of the frame file. 

 

The Frame file displays the body file and the contents file, the page file, or both. The next example shows the creation of Web-

ready Univariate procedure output using the HTML format engine with the body=, contents=, page=, and frame= options. 

 

ODS HTML body=‘ods-body.htm’ 

     contents=‘ods-contents.htm’ 

         page=‘ods-page.htm’ 

        frame=‘ods-frame.htm’  

         path=”e:/”  

         (url=none) ; 

  proc univariate data=sashelp.Cars ; 

    Title1 ‘Creating HTML Output with ODS’ ; 

  Run ; 

ODS HTML Close ; 

 

 

A snippet of the HTML output appears on the next page: 

Tip #32 – Creating PDF Output 

To share output electronically, SAS provides users with the Adobe proprietary portable document facility called PDF. The 

objective of PDF is to enable the printing of output exactly as it is seen. The significance of PDF output is that it is a great format 

for Web deployment since it is completely independent of any printer destination. To create PDF output from the UNIVARIATE 

procedure, the ODS PDF option can be specified as follows. 

 

ODS PDF FILE=‘ods-univariate.pdf’ ; 

  proc univariate data=sashelp.Cars ; 

    title1 ‘Creating PDF Output with ODS’ ; 

  run ; 

ODS PDF Close ; 

Tip #33 – Combining Output Results 

With the streaming capabilities of HTML output, results can be combined so they appear on the same screen (or page). Rather 

than having output controlled by one or more page breaks, HTML automatically displays output without page boundaries. The 

following example code illustrates combined output from the PRINT and MEANS procedures. 

 

ods html body=‘ods-body-combined.html’ 

     contents=‘ods-contents-combined.html’ 

         page=‘ods-page-combined.html’ 

        frame=‘ods-frame-combined.html’ 

         path=‘c:\sas app’ 

         (url=none) ; 

  proc print data=sashelp.Cars noobs n ; 

    title1 ‘Classic Cars Listing’ ; 

    where type in (‘SUV’) ; 

  run; 

  proc means data=sashelp.Cars ; 

    title1 ‘Summary of Classic Cars’ ; 

    class type ; 

  run ; 

ods html close ; 

 

Output from Combining Output 

Building Interactive Drill-Down Applications for SAS and Microsoft Users 

As a general rule, the best type of user interface design for transaction-based applications is a drill-down user interface 

opposed to a character-based one. It’s referred to as drill-down because a user drills down through the data, layer by layer, 

until the desired information is found. 

 

The key to building successful drill-down applications requires systems analysts and system designers to understand what users 

are trying to achieve with the data. These individuals must recognize the tasks users engage in while trying to access the desired 

information. These tasks are then translated into a series of selection criteria that users should be able to select from. 

 

Tip #34 – Building a Drill-down Application 

To simplify the process of building a graphical drill-down application in the SAS System, six easy steps are presented. 

 

1. Create a data set containing the location of the HTML link variable. 

 

2. Create HTML path with BODY and optional files. 

 

3. Create graph using HTML= option and link variable with PROC TEMPLATE. 

 

4. Produce 2-panel dashboard with PROC SGRENDER. 

 

5. Create detail list drill-down results for Pie Chart and Bar Chart. 

 

6. Use Web browser to navigate through resulting application. 

 

 

Example Drill-down Application 

In the following 2-panel drill-down application, a pie chart is used to display summary information about SUV vehicles in the left 

panel, and a vertical bar chart is used to display information about the total Price of SUV vehicles by origin (e.g., Asia, Europe 

and USA).  To display detailed information on a movie category, a user would only need to click on the desired piece of the pie 

chart or bar chart that they had interest in, as shown in the illustration below. Control would then be passed, via hyperlinks, to 

the underlying detail output that is created. 

Example Code 
The SAS coding steps used in building the 2-panel drill-down application under the Windows® platform is presented below. 

 

*************************************************************************************; 

***** Program Name: 2-panel Dashboard.SAS                                         ***; 

***** Purpose.....: Create and display a 2-panel interactive dashboard with two   ***; 

*****               categorical variables using a DATA step, PROC TEMPLATE, Pie,  ***; 

*****               Bar Chart, and PROC SGRENDER with the ODS HTML destination.   ***; 

*****                                                                             ***; 

***** Author......: Kirk Paul Lafler, Joshua Horstman and Roger Muller            ***; 

***** Date Written: 03/28/2016                                                    ***; 

***** SAS Version.: SAS 9.4                                                       ***; 

***** Input Files.: SASHELP.CARS                                                  ***; 

*****                                                                             ***; 

***** Output Files: HTML Output (17 Files)                                        ***; 

*****               2-panel-Dashboard.htm (Qty 1)                                 ***; 

*****               2-panel-Drilldown-PieChart.htm (Qty 3)                        ***; 

*****               2-panel-Drilldown-BarChart.htm (Qty 6)                        ***; 

***** Subroutines.: None                                                          ***; 

***** User-defined Formats: None                                                  ***; 

***** Macro Variables: &DSN and &MYRESULTS                                        ***; 

***** Includes....: None                                                          ***; 

***** Modification History:                                                       ***; 

*****  06/20/2017  KPL  Added Macro logic.                                        ***; 

*****  03/28/2016  KPL  Added Header information.                                 ***; 

*************************************************************************************; 

%let dsn       = sashelp.Cars ; 

%let myresults = /folders/myfolders/ ; 

libname myresult "&myresults" ; 

libname mydata '/folders/myfolders/'; 

 

*STEP 1a - Create User-defined Origin Format ; 

PROC FORMAT ; 

  VALUE $ORIGFMT 

    'Asia'    =  '2-panel-Drilldown-PieChart-Asia-Vehicles.htm' 

    'Europe'  =  '2-panel-Drilldown-PieChart-Europe-Vehicles.htm' 

    'USA'     =  '2-panel-Drilldown-PieChart-USA-Vehicles.htm' ; 

RUN ; 

 

*STEP 1b - Create User-defined Type Format ; 

PROC FORMAT ; 

  VALUE $TYPEFMT 

    'Hybrid'  =  '2-panel-Drilldown-BarChart-Hybrid-Vehicles.htm' 

    'SUV'     =  '2-panel-Drilldown-BarChart-SUV-Vehicles.htm' 

    'Sedan'   =  '2-panel-Drilldown-BarChart-Sedan-Vehicles.htm' 

    'Sports'  =  '2-panel-Drilldown-BarChart-Sports-Vehicles.htm' 

    'Truck'   =  '2-panel-Drilldown-BarChart-Truck-Vehicles.htm' 

    'Wagon'   =  '2-panel-Drilldown-BarChart-Wagon-Vehicles.htm' ; 

RUN ; 

 

* STEP 2 - Assign HTML Link to URL (Hyperlink) Variable ; 

data work.Cars_with_Hyperlinks ; 

  set &dsn ; 

  URLOriginLink = 

       cats("2-panel-Drilldown-Piechart-",origin,"-Vehicles.htm") ; 

  URLTypeLink  = 

       cats("2-panel-Drilldown-BarChart-",type,"-Vehicles.htm") ; 

run ; 

/***************************************************************************/ 

/* Step 3 - Create PieChart / BarChart Template Overlay with PROC TEMPLATE */ 

/***************************************************************************/ 

proc template ; 

  define statgraph TwoPanelDashboard ; 

    begingraph ; 

    entrytitle "2-Panel Interactive Dashboard" ; 

      layout gridded / columns=1 ; 

        layout lattice / columns=2 ; 

          cell ; 

            layout region ; 

              piechart category=Origin / url=URLOriginLink 

                                         dataskin=crisp 

                                         datatransparency=0.3 ; 

            endlayout ; 

          endcell ; 

          cell ; 

            layout overlay / width=250px  xaxisopts=(display=(label tickvalues line)) 

; 

              barchart x=Type y=MSRP / url=URLTypeLink 

                                       dataskin=crisp 

                                       datatransparency=0.3 

                                       orient=vertical 

                                       barwidth=0.8 ; 

            endlayout ; 

          endcell ; 

        endlayout ; 

      endlayout ; 

    endgraph ; 

  end ; 

run ; 

quit ; 

 

/************************************************************/ 

/* Step 4 - Produce Graphical Output from PieChart BarChart */ 

/*          Template using PROC SGRENDER                    */ 

/************************************************************/ 

ods html file='2-panel-Dashboard.htm' 

         path='/folders/myfolders/' (url=none) ; 

ods graphics / reset imagemap=on width=8in height=5in 

               imagename='2-panel-Dashboard' ; 

title1 "2-panel Dashboard" ; 

title2 "SASHELP.CARS Origin/Type" ; 

proc sgrender data=work.Cars_with_Hyperlinks 

              template=TwoPanelDashboard ; 

run ; 

quit ; 

title ; 

ods html close ; 

ods graphics off ; 

 

/***************************************************************************/ 

/* Step 5a - Create Pie Chart Drill-down Output for Each Origin Group with */ 

/*          PROC PRINT                                                     */ 

/***************************************************************************/ 

proc sql noprint ; 

  select distinct origin into :originlist separated by '~' 

    from sashelp.Cars 

      order by origin ; 

  select count(distinct origin) into 

               :numorigin separated by ' ' 

   from sashelp.Cars ; 

quit ; 

 

%macro create_drilldowns_by_origin ; 

   %do i = 1 %to &numorigin ; 

       ODS html 

           body="2-panel-Drilldown-PieChart-%SCAN(&originlist,&i,~)-Vehicles.htm" 

           path="/folders/myfolders/" (url=none) ; 

       PROC PRINT DATA=sashelp.Cars NOOBS N ; 

         TITLE "Detail Listing of %scan(&originlist,&i,~) Vehicles" ; 

         WHERE origin = "%SCAN(&originlist,&i,~)" ; 

         var make model type origin MSRP Invoice ; 

       RUN ; 

       ODS html close ; 

   %end ; 

%mend create_drilldowns_by_origin ; 

%create_drilldowns_by_origin ; 

 

/*****************************************************************************/ 

/* Step 5b - Create Bar Chart Drill-down Output for Each Type Group with     */ 

/*          PROC PRINT                                                       */ 

/*****************************************************************************/ 

proc sql noprint ; 

  select distinct type into :typelist separated by '~' 

    from sashelp.Cars 

      order by type ; 

  select count(distinct type) into 

               :numtype separated by ' ' 

   from sashelp.Cars ; 

quit ; 

 

%macro create_drilldowns_by_type ; 

   %do i = 1 %to &numtype ; 

       ODS html 

           body="2-panel-Drilldown-BarChart-%SCAN(&typelist,&i,~)-Vehicles.htm" 

           path="/folders/myfolders/" (url=none) ; 

       PROC PRINT DATA=sashelp.Cars NOOBS N ; 

         TITLE "Detail Listing of %scan(&typelist,&i,~) Vehicles" ; 

         WHERE type = "%SCAN(&typelist,&i,~)" ; 

         var make model type origin MSRP Invoice ; 

       RUN ; 

       ODS html close ; 

   %end ; 

%mend create_drilldowns_by_type ; 

%create_drilldowns_by_type ; 

Conclusion 

The time has come to turn tired-looking output listings into great looking information. The SAS
® 

Output Delivery System (ODS) 
provides new and exciting ways to improve the way output is generated. Gone are the days when the only available formatting 
choice for output was basic output listings consisting of monospace fonts printed on a white background.  
 
ODS’ built-in format engines addresses the inherent weaknesses found in traditional SAS output by enabling “quality” output to 
be produced without the need of importing it into word processors. The latest version boasts many new    format destinations 
including RTF, PDF, HTML, SAS data sets, DATA step interfaces, and EXCEL while supporting traditional Listing output as well. 
Users have a powerful and easy way to create, access and share formatted procedure and DATA step output and results to 
Microsoft Excel. 

